

A simultaneous analysis of the time-course of leukocytes and neutrophils following docetaxel administration using a semi-mechanistic myelosuppression model

Angelica Linnea Quartino · Lena E. Friberg ·
Mats O. Karlsson

Received: 6 October 2010 / Accepted: 25 November 2010 / Published online: 14 December 2010
© Springer Science+Business Media, LLC 2010

Summary Purpose To improve the predictive capacity of a semi-mechanistic myelosuppression model for neutrophils as the model have shown to over-predict the nadir of neutrophils and, secondly, to develop a model describing the time-course of leukocytes and neutrophils simultaneously. *Experimental Design* The study included 601 cancer patients treated with a 1 h infusion of docetaxel in monotherapy. A total of 3,549 pairwise observations of leukocytes and neutrophils from one treatment cycle were analyzed simultaneously in NONMEM. *Results* A basic model was developed consisting of a neutrophil and a non-neutrophil model, each with the same structure as the semi-mechanistic myelosuppression model. The leukocytes were modeled as the sum of the predicted neutrophils and non-neutrophils. The model described the time-course of the leukocytes well, but was not able to capture the nadir of the neutrophils. Hence the model was further refined and the included modifications ($p < 0.001$) in the final model are a sigmoid Emax functions for the drug effect, feedback functions on the cell maturation time in bone-marrow and an optimized number of transit compartments for each of the two cell types. *Conclusions* A joint semi-mechanistic myelosuppression model describing the time-course of leukocytes and neutrophils following docetaxel administration was developed. The data supported a more complex model compared to the previous model developed by Friberg et al. (2002), and increased the model's capacity to accurately describe the time-course of neutrophils following docetaxel therapy. The combined model also illustrates the differences between the cell types and allows prediction of neutrophil counts from leukocyte measurements.

Keywords Docetaxel · NONMEM · Myelosuppression · Neutropenia · Population pharmacokinetic-pharmacodynamic modeling · Pharmacometrics

Introduction

Pharmacokinetic-pharmacodynamic models have previously been developed to describe the time-course of leukocytes and neutrophils following chemotherapy [1–5]. Friberg et al. [5] presented a semi-physiological myelosuppression model that was developed based on leukocyte data following docetaxel, paclitaxel and etoposide data and the model was shown to adequately describe leukocyte and neutrophil data following six different chemotherapeutic drugs by estimating only four typical parameters. In addition, a slightly more complex model was presented, using an Emax model instead of a linear model to describe the drug effect, which improved the model fit for three of the compounds, one of them being docetaxel.

This model has since the original publication been successfully applied to both leukocyte and neutrophil measurements following several additional anti-cancer drugs and regimens [6–16]. It has also been shown to satisfactorily describe thrombocyte and lymphocyte measurements following chemotherapy [17–19]. Some minor modifications of the original model have been introduced in a few publications to improve the model's predictability following various types of anti-cancer drugs. Proposed alterations of the model structure include a log-linear drug effect model instead of a linear or Emax model [10], an addition of an effect delay compartment to account for the distribution of drug from the plasma to the bone marrow [10] and an addition of a neutrophil pool to describe an early increase of neutrophil count after dosage [20]. The model has further been extended to describe the

A. L. Quartino (✉) · L. E. Friberg · M. O. Karlsson
The Pharmacometrics Research Group,
Department of Pharmaceutical Biosciences, Uppsala University,
Box 591, SE-751 24 Uppsala, Sweden
e-mail: angelica.quartino@farmbio.uu.se

combined drug effect following combination therapy of anti-cancer drugs [13, 17, 18, 21–24], to incorporate the effect of administrated exogenous G-CSF [22, 25, 26] and to capture the time-course of neutrophils following peripheral blood stem-cell transplantation [26]. Additionally covariates [6, 9, 10, 20, 27–30] and inter-occasion variability [16] have been explored using the model.

We have observed that the original model does not accurately predict the time-course around the nadir for neutrophils following docetaxel treatment and hence further improvement of the model is being sought. The aims of this study were (1) to improve the predictive capacity of the myelosuppression model for neutrophils following docetaxel therapy and (2) to develop a model describing the time course of leukocytes and neutrophils simultaneously; as such an analysis may have advantages since neutrophils are the major part of the leukocytes. Such a model may also allow comparisons between neutrophils and non-neutrophils with respect to system-related parameters and drug sensitivity as well as enable prediction of neutrophil counts when only leukocytes are measured e.g. in (retrospective) academic studies where myelosuppression was not the primary aim of the study or a pooled analysis of studies where some measured neutrophils counts and some leukocyte counts. For these purposes we chose to reanalyze the data following docetaxel therapy used in Friberg et al. [5] as it is a large dataset with many samples per patient and pronounced myelosuppression. It may therefore support a more complex model of the hematological system.

Patients and methods

Patients and treatment

Data were available from 601 patients with solid tumors [breast (36%), non-small cell lung cancer (30%), and 7 other carcinomas (34%)], enrolled in 24 open controlled

phase II studies [31]. The majority of patients had visceral metastasis (82%) and almost half of all the patients (45%) had received previous chemotherapy. The patients received 75 or 100 mg/m² docetaxel in monotherapy, as a 1 h infusion every 3 weeks, but only data from the first course were available in the present analysis. A few patients had a 2–3 h infusion. Pairwise plasma leukocyte and neutrophil counts ($n=3,553$; range, 1–26 per patient) were analyzed. The pharmacokinetic data of these patients have previously been reported [31] and individual drug concentration-time profiles were generated by the published population pharmacokinetic model [32].

Model development


The semi-mechanistic myelosuppression model

The previous developed semi-mechanistic myelosuppression model (Fig. 1) consists of one proliferation compartment representing the stem cells and the proliferating precursor cells in the bone marrow. From the proliferation compartment the cells move through three transit compartments, mimicking the maturation in bone marrow, to the blood circulation compartment. The cells are eliminated from the systemic blood circulation compartment with a rate constant of k_{circ} that represent the random movement of cells into the tissue.

The rate constants (k_{tr}), describing the transfer between the different transit compartments, are set to be the same and defined as $k_{tr} = (n + 1)/MMT$, where MMT is the mean maturation time and n is the number of transitions. At steady state, the net proliferation rate (k_{prol}) is equal to k_{tr} and hence set to k_{tr} in the model. For simplicity, and because there was little information in the data on k_{circ} [5, 21], this parameter was also set to be equal to k_{tr} in the original model.

The regulation of the hematological system by endogenous growth factors, e.g. G-CSF, is included in the model

Fig. 1 Schematic presentation of the semi-mechanistic myelosuppression model by Friberg et al. [5]



as a feedback mechanism and modeled as the ratio of cell counts at baseline in the blood compartment divided by the cell counts at time t raised to a feedback factor, γ , $(Circ_0/Circ_t)^\gamma$. This function increases the production of cells in the proliferation compartment when the cell level in the blood compartment is below baseline and decrease the production when it is above baseline. The cytostatic drugs are predominantly affecting dividing cells; therefore the drug effect is coded as a reduction of the net proliferation rate that can become negative resulting in a net elimination of cells from the proliferation compartment that would capture a cytotoxic drug effect.

The original model has three system related parameters that describes the physiology: $Circ_0$ —baseline levels of cells in blood in the absence of treatment, MMT —mean maturation time of non-proliferating cell stages in the bone marrow and γ —feedback on the proliferation rate from the circulating cell count; and one parameter describing a linear drug effect: $SLOPE$.

The joint model for leukocytes and neutrophils

Two models were developed: a BASIC model and an EXTENDED model. The BASIC model was first developed to describe the time course of leukocytes and neutrophils simultaneously using the original model structure. In a second step, an EXTENDED model was developed to further improve the model's predictive capacity as we have seen that the original myelosuppression model overpredicts the nadir value and the time to nadir for neutrophils for this docetaxel data set.

The BASIC model was developed which consisted of a neutrophil and a non-neutrophil model, each with the same structure as the previously described myelosuppression model, but allowing different parameter values for neutrophils and non-neutrophils, i.e. all white blood cells except for neutrophils. The observed leukocyte count was modeled as the sum of the predicted neutrophils and non-neutrophil counts in the blood compartment. The $t_{1/2}$ in blood ($=LN(2)/k_{circ}$) was either equal to $LN(2)/k_{tr}$ (as in the published model), estimated or fixed to the literature value for neutrophils, 7 h [33, 34].

Subsequently, the BASIC model was extended to incorporate more features of the hematological system and each part of the cell chain was explored for enhancements and differences between neutrophils and non-neutrophils. For example, the following alterations to the EXTENDED model were evaluated; (1) the effect of a feedback mechanism on MMT as G-CSF is known to shorten the maturation time in the bone marrow, (2) an immediate release of neutrophils to the blood from a reserve pool, (3) a time delay of the feedback that stimulates the proliferation rate and (4) a time delay between drug concentration in

plasma and its effect on the proliferation compartment. The function describing the drug effect and the number of transit compartments were also optimized for the two structural models of the neutrophils and non-neutrophils.

Data transformation, residual error, between patient variability and correlations

An exponential model was used to describe the variability between patients in parameter values. The covariance matrix for random effects was evaluated for correlations between the parameters.

The data was Box-Cox transformed with a factor of 0.2. The value of 0.2 has in previous analyses of neutrophil data shown to give rise to weighted residuals equally distributed around zero [35, 36]. The residual error was an additive error on the Box-Cox scale. Separate residual errors for neutrophils and non-neutrophils were applied. The neutrophil residual error, applied to both neutrophils and the neutrophil component of the leukocytes thus include a common residual error of the neutrophils and leukocytes (e.g. error in sampling times, blood volume).

Data analysis and model evaluation

The data analysis was performed by non-linear-mixed-effect-modeling using the software NONMEM 6.2 [37]. The subroutine ADVAN 6 and the first-order conditional estimation (FOCE) method with interaction were used throughout the model development.

Model development was guided by the objective function value (OFV), precision in parameter estimates, graphical assessment and visual predictive checks. The OFV from NONMEM was used to differentiate between two nested models, using the log-likelihood ratio test. A decrease in OFV of 10.8 was required for the addition of one parameter to the model to be considered significant, which corresponds to a p -value of 0.001. Precision in parameter estimates was assessed by bootstrap using the PsN toolkit [38] (<http://psn.sf.net/>). Graphical judgment was performed using goodness of fit plots produced by Xpose 4 [39] (<http://xpose.sf.net/>), implemented into R version 2.6.1 (<http://www.r-project.com>). The predictive performance of the model was assessed by visual predictive check using PsN and Xpose 4. Five-hundred data sets were simulated from the model and the median and the 95% prediction interval (PI) were calculated for each of the simulated datasets. The 95% confidence interval (CI) for these calculated PI was computed and superimposed on the observed data.

The model was also evaluated in respect to its capacity to predict individual neutrophil profiles when given information of only leukocyte counts and vice versa. After

omitting all the observed neutrophil counts from all patients from the data set, the neutrophil profiles were predicted using the population parameter values and running the model without re-estimating the parameters (MAXEVAL=0). The same procedure was performed to predict leukocyte counts from neutrophil data.

Results

The basic simultaneous model

The estimated population parameters and between subject variability (BSV) of the BASIC model are presented in Table 1.

The half-life in blood was fixed to the literature value for neutrophils of 7 h [33, 34] for both neutrophils and non-neutrophils, as it has previously been shown that this type of data contain little information of this parameter [5, 21]. By separating the half-life in blood from the estimation of MMT resulted in an increase in MMT from 71 to 80 h and from 87 to 101 h for non-neutrophils and neutrophils, respectively, which are closer to MMT values reported for other anti-cancer drugs [5, 7, 9–11, 14, 15, 21, 22, 40]. A sensitivity analysis was performed which showed that the choice of half-life did not influence the estimates of the other parameters than MMT. A minor increase of the MMT parameter was seen with shorter half-lives than 7 h while a three times longer half-life resulted in a reduction of MMT of 26% for neutrophils and

Table 1 Final parameter estimates of the BASIC and EXTENDED model

		BASIC model		EXTENDED model			
		Estimate	BSV%	Estimate	RSE%	BSV%	RSE%
Neutrophils	NEU ₀ (*10 ⁹ cells/L)	5.26	39	5.07	2	40	3
	MMT (h)	101	15	102	1	11	7
	t _{1/2} blood (h)	7 FIX		7 FIX			
	γ	0.175		0.178	6		
	β			0.082	12		
	Slope (μM ⁻¹)	17.4	47				
	EC ₅₀ (μM)			1.14	8	56	11
	E _{max}			44.6	4	29	6
Non-Neutrophils	h			10.3	18		
	non-NEU ₀ (*10 ⁹ cells/L)	2.11	37	2.06	2	37	3
	MMT (h)	81.1	9	162	5	8	12
	t _{1/2} blood (h)	7 FIX		7 FIX			
	γ	0.209		0.991	8		
	Slope (μM ⁻¹)	3.48	31				
	EC ₅₀ (μM)			1.25	7	51	16
	E _{max}			99.5	7		
Residual error	h			9.52	55		
	Neutrophils	0.448	15	0.433	3	19	12
Correlations	non-Neutrophils	0.288	21	0.227	2	25	11
	non-NEU ₀ -NEU ₀ (%)	47		47		12 ^a	
	MMT _{non-NEU} -MMT _{NEU} (%)	82					
	Slope _{non-NEU} -Slope _{NEU} (%)	80					
	Residual _{non-NEU} -Residual _{NEU} (%)	54					
	Slope _{non-NEU} -MMT _{non-NEU} (%)	56					
	Slope _{non-NEU} -MMT _{NEU} (%)	8.2					
	MMT _{non-NEU} -Slope _{NEU} (%)	71					
	MMT _{NEU} -Slope _{NEU} (%)	45					
	MMT-E _{max} _{NEU} (%)			66		17 ^a	

^a Relative standard error of corresponding covariance

BSV between subject variability expressed as coefficient of variation

RSE relative standard error obtained by bootstrap (n=100)

43% for non-neutrophils. As a final step the half-life parameters were estimated using the EXTENDED model which yielded values close to 7 h for neutrophils and 2.5 h for non-neutrophils.

The visual predictive check of the BASIC model shows that the model is able to capture the time-course of the leukocytes, while a time shift is seen for the nadir of the neutrophils (Fig. 2 BASIC model).

The extended model for myelosuppression

The data supported a more complex model than the BASIC model. The final model is graphically presented in Fig. 3 and the NONMEM code is available in Appendix A.

In the original myelosuppression model the drug effect on the proliferation rate was modeled with a linear function. For docetaxel, however, an Emax function was reported to be significantly better for both the leukocyte and neutrophil data. We reinvestigated the drug-function and found that a sigmoid Emax function was significant for both neutrophils and non-neutrophils (ΔOFV -663 compared to BASIC model) and substantially improved the description of the nadir, which is seen in the VPC (Fig. 2 Sigmoid Emax model). The number of transit compartments was optimized for each of the two cell models. For neutrophils the optimal number of transit compartment was six, while one transit compartment for the non-neutrophil model was depicting the data the best (ΔOFV -141 compared to Sigmoid Emax model). The addition of transit compartments gives a sharper profile with increased time-delay before the cell counts start to fall below baseline, followed by a rapid drop in cells and a pronounced rebound. This result in an earlier and lower nadir compared with less number of transit compartments (Fig. 2 Transit compartment model). A feedback function that mimics the reduction of maturation time in bone-marrow by endogenous growth hormones was introduced (β). It was significant only on neutrophils (ΔOFV -115 compared to Transit compartment model) and resulted in a higher estimated value of MMT at the normal neutrophil levels. The introduced modifications improved the model's capability to describe the nadir value of neutrophils as shown by the visual predictive check (Fig. 2 EXTENDED model). The total difference in OFV between the BASIC and EXTENDED model was -889.

Addition of effect delay compartments between the drug plasma compartment and the proliferation compartment and between the circulating neutrophils and the feedback on the proliferation compartment were also evaluated. An addition of a neutrophil pool mimicking the reservoir of neutrophils along the blood vessels was also tested. None of these modifications improved the model fit.

The estimated population parameters and between subject variability (BSV) with the corresponding relative

standard errors (RSE) of the EXTENDED model are presented in Table 1. Based on the full variance-covariance matrix, the MMT, EC_{50} and residual error for neutrophils was 100% correlated with the corresponding parameter for non-neutrophils. The model was therefore simplified so the corresponding parameters shared the same BSV distribution, but allowing for different magnitude of BSV (i.e. completely positively correlated).


In Fig. 4 the predicted time-courses for leukocytes, neutrophils and non-neutrophils for the typical patients are shown. As depicted by the figure the neutrophils are more sensitive to docetaxel, as they reach a lower value at nadir than non-neutrophils, although the nadir occurs later for neutrophils compared to non-neutrophils. These results are in accordance with that the neutrophils are the most sensitive cells and that their maturation time show lower variability between cells as indicated by the six versus one transit compartments.

The EXTENDED model was also evaluated with respect to its ability to predict neutrophil counts from leukocyte measurements and vice versa. The epsilon-shrinkage for the neutrophils was 15%, which is reasonably low and thus it is enough information in the data to reliably estimate the individual predictions [41]. The calculation of epsilon-shrinkage is not straight forward for leukocytes as it is modeled as the sum of neutrophils and non-neutrophils. However it is expected to be in the same range as for neutrophils since the amount of data is the same for the two types of observations. The neutrophil time-course was nearly as well described by the model using only information of the leukocyte count and vice versa, as shown in Fig. 5. And for the patients who had neutropenia of grade 4, the individual predictions predicted grade 4 neutropenia in 89% of the cases.

Discussion

A joint semi-mechanistic myelosuppression model was developed that simultaneously describe the leukocyte and neutrophil time-course following docetaxel therapy. The BASIC model described the leukocyte data well, but did not capture the time-point and magnitude of the nadir for the neutrophils. This might be due to the fact that the original myelosuppression model was developed primarily using leukocyte data and with the aim to be applicable across several anti-cancer agents. Docetaxel had a more pronounced toxic effect on the neutrophils than the other five investigated drugs which may be the reason that the original model might not describe these data equally well. The modifications introduced in the EXTENDED model greatly increased the predictability of the time-course of the neutrophils. The most important modification to the model was the substitution of a linear drug effect model with a sigmoid Emax model. As the

Fig. 2 Visual predictive check for neutrophils ($\times 10^9/L$) and leukocytes ($\times 10^9/L$) versus time after dose (days) for each of the key models: the BASIC model, the Sigmoid Emax model, the Transit compartment model and the EXTENDED model. Five hundred data sets were simulated from the model and the confidence interval (*shadow area*) for the calculated median and 95% prediction interval were superimposed on the observed data (*dots*). The black lines are the corresponding median and 95% percentiles of the observed data






Fig. 3 Presentation of the final simultaneous semi-mechanistic myelosuppression model (EXTENDED model). The estimated system related parameters are: NEU_0 and $non-NEU_0$, baseline level of cells in blood; MMT , mean maturation time; γ , feedback on the proliferation rate (k_{prol})

and β , feedback on MMT . The estimated drug effect parameters are: E_{max} , maximal drug effect; EC_{50} , the concentration that gives half of maximum effect and h , the sigmoidicity factor. The parameters are estimated separately for neutrophils and non-neutrophils

authors stated in the development of the original model an Emax model was superior to a linear model for docetaxel [5]. Even though the data in this study supported a more complex drug-effect model, a few publications [7, 12, 14, 25, 29, 42] have tested an Emax model for other types of anti-cancer drugs, but have not found it to significantly improve the fit. Then again, in this study of docetaxel an Emax model showed a relative low reduction in OFV, while a sigmoid Emax model greatly improved the fit.

The neutrophil model consisted of six transit compartments that result in cellular maturation times that are more similar for all individual cells compared to when only one transit compartment is used, which was optimal for non-neutrophils. This is in line with that the non-neutrophils consist of a large variety of blood cells with different maturation processes. We were also able to characterize a second feedback mechanism by endogenous G-CSF which reduced the maturation time of neutrophils when the level of neutrophils is below baseline in blood. This is in agreement with the known reduction of post mitotic transit time of G-CSF [43–45].

The estimated parameters of the BASIC model for neutrophils differ from the published myelosuppression model by Friberg et al. [5] The SLOPE parameter is higher in our analysis possibly due to that the estimation method FOCE was used for all parameters instead of the hybrid

method, where the FO method was used to estimate MMT and SLOPE. The FO method is known to induce bias in parameters and therefore is our estimate of the SLOPE value likely more accurate. The MMT increased to 101 h by fixing the half-life in blood to 7 h as opposed to fixing it to be equal to $LN(2)/k_{tr}$ (as in the previously published model). Thereby the estimate of MMT approached the estimated values for the post mitotic transit time for neutrophils for other anti-cancer drugs [5, 7, 9–11, 14–16, 21, 22, 40]. However this is lower than the reported value





Fig. 4 Prediction of the typical time-course of neutrophils and non-neutrophils after docetaxel administration (100 mg/m^2)

Fig. 5 Observations versus model predictions for neutrophils (**a**) and total leukocyte counts (**b**). The two left panels show the individual predictions and the population predictions from the final (EXTENDED) model and the right panels are the individual predictions of neutrophils based on leukocyte data only and predictions of leukocytes based on neutrophil data only, respectively



of 158 h in untreated healthy volunteers [46], which may be a consequence of that some of the feedback from growth factors released during the treatment cycle is incorporated in the MMT parameter. Another reason may be that the majority of patients in this study have received previous chemotherapy treatment and thus they may already have an augmented granulopoiesis due to increased levels of growth factors such as G-CSF at the start of this study.

The simultaneous analysis quantified the differences in the time-course and drug sensitivity between the neutrophils and non-neutrophils. In addition it showed that the neutrophil fraction of leukocytes is changing over time. These differences need to be considered, hence prediction of neutrophil count if not measured is important, e.g. in academic settings where neutrophils are not routinely measured. Even if a simultaneous analysis using this large dataset contributed little to the improved description of the neutrophil nadir, a smaller dataset with sparse sampling will likely benefit from a simultaneous analysis to support the model, especially if the

leukocyte measurements are richer than the neutrophil measurements or if they occur at different time points.

In conclusion, a simultaneous analysis of the time-course of neutrophils and leukocytes was successfully performed. The data supported a more complex model for the hematological system compared to the previous model developed by Friberg et al. [5] and yielded more precise predictions of the time-course of the neutrophil counts. The model shows good simulation properties and allows utilization of all data available. A combined model may be useful in illustrating the differences between the cell types and allows prediction of neutrophil counts from leukocyte measurements alone.

Acknowledgment We would like to thank Dr Rene Bruno for providing the data. This research was supported by the Swedish Cancer Society. Lena Friberg received a grant from Knut and Alice Wallenberg foundation, Sweden.

Conflict of interest The authors declare no conflicts of interest.

Appendix A

```

$PROB FINAL EXTENDED MYELOSUPPRESSION MODEL
$INPUT ID TIME DV FLG L2 CP EVID

$DATA data.csv IGN=@
$$SUBS ADVAN6 TOL=7

$MODEL

COMP=(CIRCIN) ;1 Systemic circulation - non-Neutrophils
COMP=(STEMIN) ;2
COMP=(TRANS1IN) ;3

COMP=(CIRCN) ;4 Systemic circulation - Neutrophils
COMP=(STEMN) ;5
COMP=(TRANS1N) ;6
COMP=(TRANS2N) ;7
COMP=(TRANS3N) ;8
COMP=(TRANS4N) ;9
COMP=(TRANS5N) ;10
COMP=(TRANS6N) ;11

$PK
"FIRST
" COMMON /PRCOMG/ IDUM1,IDUM2,IMAX,IDUM4,IDUM5
" INTEGER IDUM1,IDUM2,IMAX,IDUM4,IDUM5
" IMAX=70000000

;----- Pharmacokinetics -----
CO = CP ;predicted plasma concentrations of docetaxel in ng/L

;----- Non-Neutrophils-----
BASNN = THETA(1)*EXP(ETA(1))
MMTNN = THETA(2)*EXP(ETA(3))
KNN = (2/MMTNN)
KENN = LOG(2)/THETA(3)
GAMNN = THETA(4)
E50NN = THETA(5)*808/1000*EXP(ETA(5)) ;convert from μmol/L to ng/L
EMXNN = THETA(6)
HILNN = THETA(7)

A_INITIAL(1) = BASNN
A_INITIAL(2) = (KENN*BASNN)/KNN
A_INITIAL(3) = (KENN*BASNN)/KNN

;----- Neutrophils -----
BASN = THETA(9)*EXP(ETA(2))
MMTN = THETA(10)*EXP(THETA(18)*ETA(3))
KN = (7/MMTN)
KEN = LOG(2)/THETA(11)
GAMN = THETA(12)
BETN = THETA(13)

E50N = THETA(14)*808/1000*EXP(THETA(19) * ETA(5)) ;convert from μmol/L to ng/L
EMXN = THETA(15)*EXP(ETA(4))
HILN = THETA(16)

A_INITIAL(4) = BASN
A_INITIAL(5) = (KEN*BASN)/KN
A_INITIAL(6) = (KEN*BASN)/KN
A_INITIAL(7) = (KEN*BASN)/KN
A_INITIAL(8) = (KEN*BASN)/KN
A_INITIAL(9) = (KEN*BASN)/KN
A_INITIAL(10) = (KEN*BASN)/KN
A_INITIAL(11) = (KEN*BASN)/KN

```

```

$DES

;----PD-----
DRUNN = (EMXNN*CO**HILNN)/(E50NN**HILNN+CO**HILNN)
DRUN  = (EMXN*CO**HILN)/(E50N**HILN+CO**HILN)

;----Physiology -----
;----- Non-Neutrophils -----
FNN = (BASNN/0.0001)**GAMNN
IF (A(1).GT.0.0001) FNN = (BASNN/A(1))**GAMNN ;feedback on proliferation rate

DADT(1) = KNN*A(3) - KENN*A(1) ;circulating non-Neutrophils
DADT(2) = KNN*A(2)*(1-DRUNN)*FNN - KNN*A(2) ;proliferation compartment
DADT(3) = KNN*A(2) - KNN*A(3) ;transit compartment

;-----Neutrophils-----
FBMN = (BASN/0.0001)**BETN
IF (A(4).GT.0.0001) FBMN = (BASN/A(4))**BETN ;feedback on maturation time (MMT)

FN = (BASN/0.0001)**GAMN
IF (A(4).GT.0.0001) FN = (BASN/A(4))**GAMN ;feedback on proliferation rate

DADT(4) = KN*A(11)*FBMN - KEN*A(4) ;circulating Neutrophils
DADT(5) = KN*A(5) *(1-DRUN)*FN - KN*A(5)*FBMN ;proliferation compartment
DADT(6) = KN*A(5) *FBMN - KN*A(6) *FBMN ;transit compartment
DADT(7) = KN*A(6) *FBMN - KN*A(7) *FBMN ;transit compartment
DADT(8) = KN*A(7) *FBMN - KN*A(8) *FBMN ;transit compartment
DADT(9) = KN*A(8) *FBMN - KN*A(9) *FBMN ;transit compartment
DADT(10) = KN*A(9) *FBMN - KN*A(10)*FBMN ;transit compartment
DADT(11) = KN*A(10)*FBMN - KN*A(11)*FBMN ;transit compartment

;-----
$error (ONLY OBSERVATIONS)

;---- WBC: FLG=2 and NEU: FLG=1 -----
NNEU = A(1) ; non-Neutrophils
NEU = A(4) ; Neutrophils
WBC = NEU+INEU ; Leukocytes

IPRED = (WBC**0.2-1)/0.2 ;convert WBC to Box-Cox scale
IF (FLG.EQ.1) IPRED = (NEU**0.2-1)/0.2 ;convert NEU to Box-Cox scale

;---- Residual error model -----
W1 = THETA(8) *EXP(ETA(6)) ;additive residual error on Box-Cox scale
W2 = THETA(17)*EXP(THETA(20)*ETA(6)) ;additive residual error on Box-Cox scale

; The residual error is additive on Box-Cox scale for both non-NEU and NEU.
; The predictions are therefore Box-Cox transformed before the residual error is added.

NNEUE = ((NNEU**0.2-1)/0.2) + W1*EPS(1) ; non-NEU with res. error on Box-Cox scale
NEUE = ((NEU**0.2-1)/0.2) + W2*EPS(2) ; NEU with res. error on Box-Cox scale

; WBC is the sum of non-NEU and NEU on untransformed scale. Why they are converted back to
; untransformed scale before WBC is calculated (WBC1).
; Thereafter WBC is Box-Cox transformed as it is in the dataset (WBCE).

WBC1 = ((NNEUE*0.2+1)**(1/0.2))+((NEUE*0.2+1)**(1/0.2))
WBCE = ((WBC1**0.2-1)/0.2) ; WBC with residual error on Box-Cox scale

Y = WBCE ; WBC on Box-Cox scale (as in the dataset)
IF(FLG.EQ.1) Y = NEUE ; NEU on Box-Cox scale (as in the dataset)

```

```

-----
$THETA (0, 2.05) ;1 BASE non-neutrophils
$THETA (0, 158) ;2 MMT non-neutrophils
$THETA (7 FIX) ;3 T1/2 non-neutrophils
$THETA (0, 0.971) ;4 GAMMA non-neutrophils
$THETA (0, 1.11) ;5 EC50 non-neutrophils
$THETA (0, 93.4) ;6 Emax non-neutrophils
$THETA (0, 9.53) ;7 HILL non-neutrophils
$THETA (0, 0.282) ;8 Residual error non-neutrophils

$THETA (0, 5.07) ;9 BASE Neutrophils
$THETA (0, 103) ;10 MMT Neutrophils
$THETA (7 FIX) ;11 T1/2 Neutrophils
$THETA (0, 0.178) ;12 GAMMA Neutrophils
$THETA (0, 0.0816) ;13 BETA Neutrophils
$THETA (0, 1.34) ;14 EC50 Neutrophils
$THETA (0, 45.7) ;15 Emax Neutrophils
$THETA (0, 11.9) ;16 HILL Neutrophils
$THETA (0, 0.44) ;17 Residual error Neutrophils

$THETA (0, 2) ;18 IIV MMT correction factor for neutrophils
$THETA (0, 1.5) ;19 IIV EC50 correction factor for neutrophils
$THETA (0, 0.5) ;20 IIV Residual error correction factor for neutrophils

-----

$OMEGA BLOCK (2)
0.135 ;1 IIV BASE non-neutrophils
0.068 0.161 ;2 IIV BASE Neutrophils
$OMEGA BLOCK (2)
0.08 ;3 IIV MMT
0.02 0.09 ;4 IIV EMAX Neutrophils
$OMEGA 0.4 ;5 IIV EC50
$OMEGA 0.06 ;6 IIV residual error

$SIGMA 1 FIX ;non-neutrophils
$SIGMA 1 FIX ;neutrophils

-----

$EST METHOD=1 INTER MAX=9999 PRINT=1
$COV PRINT=E

```

References

- Zamboni WC, D'Argenio DZ, Stewart CF, MacVittie T, Delauter BJ, Farese AM, Potter DM, Kubat NM, Tubergen D, Egorin MJ (2001) Pharmacodynamic model of topotecan-induced time course of neutropenia. *Clin Cancer Res* 7(8):2301–2308
- Panetta JC, Kirstein MN, Gajjar AJ, Nair G, Fouladi M, Stewart CF (2003) A mechanistic mathematical model of temozolomide myelosuppression in children with high-grade gliomas. *Math Biosci* 186(1):29–41
- Bulitta JB, Zhao P, Arnold RD, Kessler DR, Daifuku R, Pratt J, Luciano G, Hanauske AR, Gelderblom H, Awada A, Jusko WJ (2009) Multiple-pool cell lifespan models for neutropenia to assess the population pharmacodynamics of unbound paclitaxel from two formulations in cancer patients. *Cancer Chemother Pharmacol* 63(6):1035–1048
- Minami H, Sasaki Y, Saijo N, Ohtsu T, Fujii H, Igarashi T, Itoh K (1998) Indirect-response model for the time course of leukopenia with anticancer drugs. *Clin Pharmacol Ther* 64(5):511–521. doi:10.1016/S0009-9236(98)90134-5
- Friberg LE, Henningsson A, Maas H, Nguyen L, Karlsson MO (2002) Model of chemotherapy-induced myelosuppression with parameter consistency across drugs. *J Clin Oncol* 20(24):4713–4721. doi:10.1200/JCO.2002.02.140
- Léger F, Loos WJ, Bugat R, Mathijssen RHJ, Goffinet M, Verweij J, Sparreboom A, Chatelut E (2004) Mechanism-based models for topotecan-induced neutropenia. *Clin Pharmacol Ther* 76(6):567–578
- Troconiz IF, Garrido MJ, Segura C, Cendros JM, Principe P, Peraire C, Obach R (2006) Phase I dose-finding study and a pharmacokinetic/pharmacodynamic analysis of the neutropenic response of intravenous diflomotecan in patients with advanced malignant tumours. *Cancer Chemother Pharmacol* 57(6):727–735
- Van Kesteren C, Zandvliet AS, Karlsson MO, Mathôt RAA, Punt CJA, Armand JP, Raymond E, Huitema ADR, Dittrich C, Dumez H, Roche HH, Droz JP, Ravic M, Yule SM, Wanders J, Beijnen JH, Fumoleau P, Schellens JHM (2005) Semi-physiological model describing the hematological toxicity of the anti-cancer agent indisulam. *Investig New Drugs* 23(3):225–234
- Latz JE, Karlsson MO, Rusthoven JJ, Ghosh A, Johnson RD (2006) A semimechanistic-physiologic population pharmacokinetic/pharmacodynamic model of chemotherapy-induced myelosuppression. *Clin Cancer Res* 12(12):3613–3624

- dynamic model for neutropenia following pemetrexed therapy. *Cancer Chemother Pharmacol* 57(4):412–426
10. Hing J, Perez-Ruixo JJ, Stuyckens K, Soto-Matos A, Lopez-Lazaro L, Zannikos P (2008) Mechanism-based pharmacokinetic/pharmacodynamic meta-analysis of trabectedin (et-743, yondelis) induced neutropenia. *Clin Pharmacol Ther* 83(1):130–143. doi:10.1038/sj.clpt.6100259
 11. Brain EGC, Rezaei K, Lokiec F, Gutierrez M, Urien S (2008) Population pharmacokinetics and exploratory pharmacodynamics of ifosfamide according to continuous or short infusion schedules: an $n=1$ randomized study. *Br J Clin Pharmacol* 65(4):607–610
 12. Kathman SJ, Williams DH, Hodge JP, Dar M (2007) A bayesian population pk-pd model of ispinesib-induced myelosuppression. *Clin Pharmacol Ther* 81(1):88–94
 13. Kathman SJ, Williams DH, Hodge JP, Dar M (2009) A bayesian population pk-pd model for ispinesib/docetaxel combination-induced myelosuppression. *Cancer Chemother Pharmacol* 63(3):469–476
 14. Soto E, Staab A, Tillmann C, Trommeshauser D, Fritsch H, Munzert G, Troconiz IF (2010) Semi-mechanistic population pharmacokinetic/pharmacodynamic model for neutropenia following therapy with the Plk-1 inhibitor BI 2536 and its application in clinical development. *Cancer Chemother Pharmacol* 66(4):785–95
 15. Ng CM, Patnaik A, Beeram M, Lin CC, Takimoto CH (2010) Mechanism-based pharmacokinetic/pharmacodynamic model for troxacitabine-induced neutropenia in cancer patients. *Cancer Chemother Pharmacol*. doi:10.1007/s00280-010-1393-y
 16. Hansson EK, Wallin JE, Lindman H, Sandström M, Karlsson MO, Friberg LE (2010) Limited inter-occasion variability in relation to inter-individual variability in chemotherapy-induced myelosuppression. *Cancer Chemother Pharmacol* 65(5):839–848
 17. Zandvliet AS, Siegel-Lakshai WS, Beijnen JH, Copalu W, Etienne-Grimaldi MC, Milano G, Schellens JHM, Huitema ADR (2008) Pk/pd model of indisulam and capecitabine: interaction causes excessive myelosuppression. *Clin Pharmacol Ther* 83(6):829–839. doi:10.1038/sj.clpt.6100344
 18. Zandvliet AS, Schellens JHM, Ditttrich C, Wanders J, Beijnen JH, Huitema ADR (2008) Population pharmacokinetic and pharmacodynamic analysis to support treatment optimization of combination chemotherapy with indisulam and carboplatin. *Br J Clin Pharmacol* 66(4):485–497. doi:10.1111/j.1365-2125.2008.03230.x
 19. Segura C, Bandres E, Troconiz IF, Garcia-Foncillas J, Sayar O, Dios-Vieitez C, Renedo MJ, Garrido MJ (2004) Hematological response of topotecan in tumor-bearing rats: modeling of the time course of different cellular populations. *Pharm Res* 21(4):567–573
 20. Ozawa K, Minami H, Sato H (2007) Population pharmacokinetic and pharmacodynamic analysis for time courses of docetaxel-induced neutropenia in japanese cancer patients. *Cancer Sci* 98(12):1985–1992
 21. Sandstrom M, Lindman H, Nygren P, Lidbrink E, Bergh J, Karlsson MO (2005) Model describing the relationship between pharmacokinetics and hematologic toxicity of the epirubicin-docetaxel regimen in breast cancer patients. *J Clin Oncol* 23(3):413–421
 22. Sandstrom M, Lindman H, Nygren P, Johansson M, Bergh J, Karlsson MO (2006) Population analysis of the pharmacokinetics and the haematological toxicity of the fluorouracil-epirubicin-cyclophosphamide regimen in breast cancer patients. *Cancer Chemother Pharmacol* 58(2):143–156
 23. Joerger M, Huitema ADR, Richel DJ, Ditttrich C, Pavlidis N, Briasoulis E, Vermorken JB, Strocchi E, Martoni A, Sorio R, Sleeboom HP, Izquierdo MA, Jodrell DI, Féty R, De Bruijn E, Hempel G, Karlsson M, Tranchand B, Schrijvers AHGJ, Twelves C, Beijnen JH, Schellens JHM (2007) Population pharmacokinetics and pharmacodynamics of doxorubicin and cyclophosphamide in breast cancer patients: a study by the eortc-pamm-nddg. *Clin Pharmacokinetics* 46(12):1051–1068
 24. Joerger M, Huitema ADR, Richel DJ, Ditttrich C, Pavlidis N, Briasoulis E, Vermorken JB, Strocchi E, Martoni A, Sorio R, Sleeboom HP, Izquierdo MA, Jodrell DI, Calvert H, Boddy AV, Hollema H, Féty R, Van Der Vijgh WJF, Hempel G, Chatelut E, Karlsson M, Wilkins J, Tranchand B, Schrijvers AHGJ, Twelves C, Beijnen JH, Schellens JHM (2007) Population pharmacokinetics and pharmacodynamics of paclitaxel and carboplatin in ovarian cancer patients: a study by the european organization for research and treatment of cancer-pharmacology and molecular mechanisms group and new drug development group. *Clin Cancer Res* 13(21):6410–6418. doi:10.1158/1078-0432.CCR-07-0064
 25. Zandvliet AS, Schellens JHM, Copalu W, Beijnen JH, Huitema ADR (2009) Covariate-based dose individualization of the cytotoxic drug indisulam to reduce the risk of severe myelosuppression. *J Pharmacokinetics Pharmacodyn* 36(1):39–62
 26. Ramon-Lopez A, Nalda-Molina R, Valenzuela B, Perez-Ruixo JJ (2009) Semi-mechanistic model for neutropenia after high dose of chemotherapy in breast cancer patients. *Pharm Res* 26(8):1952–1962
 27. Latz JE, Schneck KL, Nakagawa K, Miller MA, Takimoto CH (2009) Population pharmacokinetic/pharmacodynamic analyses of pemetrexed and neutropenia: effect of vitamin supplementation and differences between japanese and western patients. *Clin Cancer Res* 15(1):346–354
 28. Puisset F, Alexandre J, Treluyer JM, Raoul V, Roche H, Goldwasser F, Chatelut E (2007) Clinical pharmacodynamic factors in docetaxel toxicity. *Br J Cancer* 97(3):290–296
 29. Joerger M, Huitema ADR, Huizing MT, Willemsse PHB, De Graeff A, Rosing H, Schellens JHM, Beijnen JH, Vermorken JB (2007) Safety and pharmacology of paclitaxel in patients with impaired liver function: a population pharmacokinetic-pharmacodynamic study. *Br J Clin Pharmacol* 64(5):622–633
 30. Kloft C, Wallin J, Henningsson A, Chatelut E, Karlsson MO (2006) Population pharmacokinetic-pharmacodynamic model for neutropenia with patient subgroup identification: comparison across anticancer drugs. *Clin Cancer Res* 12(18):5481–5490
 31. Bruno R, Hille D, Riva A, Vivier N, ten Bokkel Huinink WW, van Oosterom AT, Kaye SB, Verweij J, Fossella FV, Valero V, Rigas JR, Seidman AD, Chevallier B, Fumoleau P, Burris HA, Ravdin PM, Sheiner LB (1998) Population pharmacokinetics/pharmacodynamics of docetaxel in phase ii studies in patients with cancer. *J Clin Oncol* 16(1):187–196
 32. Bruno R, Vivier N, Vergniol JC, De Phillips SL, Montay G, Sheiner LB (1996) A population pharmacokinetic model for docetaxel (taxotere): model building and validation. *J Pharmacokinetics Biopharm* 24(2):153–172
 33. Ganong W (2010) Review of medical physiology, 23rd edition. The McGraw Hill companies, U.S ISBN: 0-07-144040-2
 34. Smith CW (2010) Williams hematology 8th edition. The McGraw Hill companies. U.S Book ISBN: 978-0-07-162144-1
 35. Friberg LE, Brindley CJ, Karlsson MO, Devlin AJ (2000) Models of schedule dependent haematological toxicity of 2'-deoxy-2'-methylidenecytidine (dmdec). *Eur J Clin Pharmacol* 56(8):567–574
 36. Karlsson MO, Port RE, Ratain MJ, Sheiner LB (1995) A population model for the leukopenic effect of etoposide. *Clin Pharmacol Ther* 57(3):325–334
 37. Beal S, Sheiner L, Boeckmann A, Bauer R (1989–2009) Nonmem user's guides. Elliot City, MD, USA
 38. Lindbom L, Pihlgren P, Jonsson EN (2005) Psn-toolkit—a collection of computer intensive statistical methods for non-linear mixed effect modeling using nonmem. *Comput Methods Programs Biomed* 79(3):241–257
 39. Jonsson EN, Karlsson MO (1999) Xpose—an s-plus based population pharmacokinetic/pharmacodynamic model building aid for nonmem. *Comput Methods Programs Biomed* 58(1):51–64

40. Leger F, Loos WJ, Bugat R, Mathijssen RH, Goffinet M, Verweij J, Sparreboom A, Chatelut E (2004) Mechanism-based models for topotecan-induced neutropenia. *Clin Pharmacol Ther* 76(6):567–578
41. Karlsson MO, Savic RM (2007) Diagnosing model diagnostics. *Clin Pharmacol Ther* 82(1):17–20. doi:10.1038/sj.clpt.6100241
42. van Kesteren C, Zandvliet AS, Karlsson MO, Mathot RA, Punt CJ, Armand JP, Raymond E, Huitema AD, Dittrich C, Dumez H, Roche HH, Droz JP, Ravic M, Yule SM, Wanders J, Beijnen JH, Fumoleau P, Schellens JH (2005) Semi-physiological model describing the hematological toxicity of the anti-cancer agent indisulam. *Invest New Drugs* 23(3):225–234
43. Lord BI, Bronchud MH, Owens S, Chang J, Howell A, Souza L, Dexter TM (1989) The kinetics of human granulopoiesis following treatment with granulocyte colony-stimulating factor in vivo. *Proc Natl Acad Sci USA* 86(23):9499–9503
44. Schmitz S, Franke H, Brusis J, Wichmann HE (1993) Quantification of the cell kinetic effects of g-csf using a model of human granulopoiesis. *Exp Hematol* 21(6):755–760
45. Price TH, Chatta GS, Dale DC (1996) Effect of recombinant granulocyte colony-stimulating factor on neutrophil kinetics in normal young and elderly humans. *Blood* 88(1):335–340
46. Dancy JT, Deubelbeiss KA, Harker LA, Finch CA (1976) Neutrophil kinetics in man. *J Clin Invest* 58(3):705–715